We study a bipendulum, namely a system composed of a bar where two pendula are fixed, one of length $l1$ and one of length $l2$. See J.-F. Pommaret, *Partial Differential Control Theory*, Kluwer, 2001, p. 569.

```
> with(Ore_algebra):
> with(OreModules):
```

The appropriate Ore algebra for this example is the Weyl algebra $Alg = A_1$, where D is the differential operator w.r.t. time t:

```
> Alg := DefineOreAlgebra(diff=[D,t], polynom=[t], comm=[g, l1, l2]):
```

Note that we have to declare all constants appearing in the system equations (the gravitational constant g, and the lengths $l1$, $l2$) as variables that "comm"ute with D and t. Next we enter the system matrix:

```
> R := evalm([[D^2+g/l1, 0, -g/l1], [0, D^2+g/l2, -g/l2]]);
```

$$R := \begin{bmatrix} \mathrm{D}^2 + \frac{g}{l1} & 0 & -\frac{g}{l1} \\ 0 & \mathrm{D}^2 + \frac{g}{l2} & -\frac{g}{l2} \end{bmatrix}$$

In terms of equations, the bipendulum is defined by:

```
> ApplyMatrix(R, [x1(t),x2(t),u(t)], Alg) = evalm([[0],[0]]);
```

$$\begin{bmatrix} \frac{g\,\mathrm{x1}(t)}{l1} + (\frac{d^2}{dt^2}\,\mathrm{x1}(t)) - \frac{g\,\mathrm{u}(t)}{l1} \\ \frac{g\,\mathrm{x2}(t)}{l2} + (\frac{d^2}{dt^2}\,\mathrm{x2}(t)) - \frac{g\,\mathrm{u}(t)}{l2} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \end{bmatrix}$$

We compute the formal adjoint of R:

```
> R_adj := Involution(R, Alg);
```

$$R\_adj := \begin{bmatrix} \mathrm{D}^2 + \frac{g}{l1} & 0 \\ 0 & \mathrm{D}^2 + \frac{g}{l2} \\ -\frac{g}{l1} & -\frac{g}{l2} \end{bmatrix}$$

By computing the first extension module ext^1 with values in Alg of the module associated with the formal adjoint of R, we check controllability and, equivalently, parametrizability of the bipendulum:

```
> Ext := Exti(R_adj, Alg, 1);
```

$$Ext := \left[\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} \mathrm{D}^2\,l1 + g & 0 & -g \\ 0 & \mathrm{D}^2\,l2 + g & -g \end{bmatrix}, \begin{bmatrix} l2\,g\,\mathrm{D}^2 + g^2 \\ g^2 + \mathrm{D}^2\,l1\,g \\ l2\,g\,\mathrm{D}^2 + l2\,\mathrm{D}^4\,l1 + \mathrm{D}^2\,l1\,g + g^2 \end{bmatrix} \right]$$

From the output of *Exti*, we see that the system is *generically* controllable because $Ext[1]$ is the identity matrix which means that there are no torsion elements in the Alg-module M which is associated with the system. The interpretation of this structural fact is that the system has no autonomous elements *in the generic case*. There may be configurations of the constants g, $l1$, $l2$, in which the bipendulum is not controllable. We shall actually find the only configuration where it is not controllable below. Let us write down the generic parametrization $Ext[3]$ in a more familiar way with a free function ξ_1:

```
> P := Parametrization(R, Alg);
```

$$P := \begin{bmatrix} g\,(g\,\xi_1(t) + \mathit{l2}\,\%1) \\ g\,(g\,\xi_1(t) + \mathit{l1}\,\%1) \\ g^2\,\xi_1(t) + g\,\mathit{l2}\,\%1 + g\,\mathit{l1}\,\%1 + \mathit{l1}\,\mathit{l2}\,(\frac{d^4}{dt^4}\,\xi_1(t)) \end{bmatrix}$$

$$\%1 := \frac{d^2}{dt^2}\,\xi_1(t)$$

Therefore, all solutions of the system are parametrized by P, i.e.,

$$R\,(x1 : x2 : u)^T = 0 \iff (x1 : x2 : u)^T = Ext[3]\,\xi_1.$$

Since *Alg* is a principal ideal domain, torsion-free modules are free modules. Hence, the bipendulum is also generically a flat system. A flat output of the system can be obtained by computing as a left-inverse of the parametrization *Ext*[3]:

```
> S := LeftInverse(Ext[3], Alg);
```

$$S := \begin{bmatrix} \frac{\mathit{l1}}{g^2\,(-\mathit{l2} + \mathit{l1})} & -\frac{\mathit{l2}}{g^2\,(-\mathit{l2} + \mathit{l1})} & 0 \end{bmatrix}$$

Then, a flat output of the system is defined by $\xi_1 = S\ (\mathit{x1} : \mathit{x2} : u)^T$, namely:

```
> ApplyMatrix(S, [x1(t),x2(t),u(t)], Alg);
```

$$\begin{bmatrix} \frac{\mathit{l1}\,\mathrm{x1}(t)}{g^2\,(-\mathit{l2} + \mathit{l1})} - \frac{\mathit{l2}\,\mathrm{x2}(t)}{g^2\,(-\mathit{l2} + \mathit{l1})} \end{bmatrix}$$

We remark that this flat output is defined only if $\mathit{l1} - \mathit{l2} \neq 0$. Moreover, $\mathit{l1} = \mathit{l2}$ describes the only case in which the bipendulum may be uncontrollable.

Let us compute the Brunovský canonical form of the system in the case where $\mathit{l1} \neq \mathit{l2}$.

```
> B := Brunovsky(R, Alg);
```

$$B := \begin{bmatrix} \frac{\mathit{l1}}{g^2\,(-\mathit{l2} + \mathit{l1})} & -\frac{\mathit{l2}}{g^2\,(-\mathit{l2} + \mathit{l1})} & 0 \\ \frac{\mathrm{D}\,\mathit{l1}}{g^2\,(-\mathit{l2} + \mathit{l1})} & -\frac{\mathrm{D}\,\mathit{l2}}{g^2\,(-\mathit{l2} + \mathit{l1})} & 0 \\ -\frac{1}{g\,(-\mathit{l2} + \mathit{l1})} & \frac{1}{g\,(-\mathit{l2} + \mathit{l1})} & 0 \\ -\frac{\mathrm{D}}{g\,(-\mathit{l2} + \mathit{l1})} & \frac{\mathrm{D}}{g\,(-\mathit{l2} + \mathit{l1})} & 0 \\ \frac{1}{(-\mathit{l2} + \mathit{l1})\,\mathit{l1}} & -\frac{1}{(-\mathit{l2} + \mathit{l1})\,\mathit{l2}} & \frac{1}{\mathit{l1}\,\mathit{l2}} \end{bmatrix}$$

In other words, we have the following transformation between the system variables *x1*, *x2* and u and the Brunovský variables $z[i]$ and v:

```
> evalm([seq([z[i](t)],i=1..4),[v(t)]])=ApplyMatrix(B, [x1(t),x2(t),u(t)], Alg);
```

$$\left[\begin{array}{c} z_1(t) \\ z_2(t) \\ z_3(t) \\ z_4(t) \\ \mathrm{v}(t) \end{array}\right] = \left[\begin{array}{c} \frac{l1\,\mathrm{x1}(t)}{g^2\,(-l2+l1)} - \frac{l2\,\mathrm{x2}(t)}{g^2\,(-l2+l1)} \\ \frac{l1\,(\frac{d}{dt}\,\mathrm{x1}(t))}{g^2\,(-l2+l1)} - \frac{l2\,(\frac{d}{dt}\,\mathrm{x2}(t))}{g^2\,(-l2+l1)} \\ -\frac{\mathrm{x1}(t)}{g\,(-l2+l1)} + \frac{\mathrm{x2}(t)}{g\,(-l2+l1)} \\ -\frac{\frac{d}{dt}\,\mathrm{x1}(t)}{g\,(-l2+l1)} + \frac{\frac{d}{dt}\,\mathrm{x2}(t)}{g\,(-l2+l1)} \\ \frac{\mathrm{x1}(t)}{(-l2+l1)\,l1} - \frac{\mathrm{x2}(t)}{(-l2+l1)\,l2} + \frac{\mathrm{u}(t)}{l1\;l2} \end{array}\right]$$

Let us check that the new variables $z[i]$ and v satisfy a Brunovský canonical form:

```
> F := Elimination(linalg[stackmatrix](B, R), [x1,x2,u],
> [seq(z[i],i=1..4),v,0,0], Alg):
> ApplyMatrix(F[1], [x1(t),x2(t),u(t)], Alg)=
> ApplyMatrix(F[2], [seq(z[i](t),i=1..4),v(t)], Alg);
```

$$\left[\begin{array}{c} 0 \\ 0 \\ 0 \\ 0 \\ \mathrm{u}(t) \\ \mathrm{x2}(t) \\ \mathrm{x1}(t) \end{array}\right] = \left[\begin{array}{c} -(\frac{d}{dt}\,z_4(t)) + \mathrm{v}(t) \\ -(\frac{d}{dt}\,z_3(t)) + z_4(t) \\ -(\frac{d}{dt}\,z_2(t)) + z_3(t) \\ -(\frac{d}{dt}\,z_1(t)) + z_2(t) \\ g^2\,z_1(t) + (g\,l2 + g\,l1)\,z_3(t) + l1\;l2\,\mathrm{v}(t) \\ g^2\,z_1(t) + g\,l1\,z_3(t) \\ g^2\,z_1(t) + g\,l2\,z_3(t) \end{array}\right]$$

We now turn to the case where the lengths of the pendula are equal:

```
> R_mod := subs(l2=l1, evalm(R));
```

$$R\_mod := \left[\begin{array}{ccc} \mathrm{D}^2 + \frac{g}{l1} & 0 & -\frac{g}{l1} \\ 0 & \mathrm{D}^2 + \frac{g}{l1} & -\frac{g}{l1} \end{array}\right]$$

```
> Ext_mod := Exti(Involution(R_mod, Alg), Alg, 1);
```

$$Ext\_mod := \left[\left[\begin{array}{cc} \mathrm{D}^2\,l1 + g & 0 \\ 0 & 1 \end{array}\right], \left[\begin{array}{ccc} 1 & -1 & 0 \\ 0 & \mathrm{D}^2\,l1 + g & -g \end{array}\right], \left[\begin{array}{c} g \\ g \\ \mathrm{D}^2\,l1 + g \end{array}\right]\right]$$

The computation of the first extension module ext^1 with values in *Alg* of the module defined by the formal adjoint of *R_mod* gives the torsion submodule $\mathrm{t}(M)$ of M: it is generated by the row r of *Ext_mod*[2] which corresponds to the row with entry $l1\,\mathrm{D}^2 + g$ in *Ext_mod*[1]. This means that $(l1\,D^2 + g)\,r = 0$ in M, and the difference *x1 - x2* of the positions of the pendula (relative to the bar) is an autonomous element of the system. We conclude that the bipendulum is controllable if and only if $l1 \neq l2$.

Let us point out that we can directly obtain the torsion elements of M as follows:

```
> TorsionElements(R_mod, [x1(t),x2(t),u(t)], Alg);
```

$$[\left[\; g\,\theta_1(t) + l1\,(\frac{d^2}{dt^2}\,\theta_1(t)) = 0 \;\right], \left[\; \theta_1(t) = \mathrm{x1}(t) - \mathrm{x2}(t) \;\right]]$$

We can explicitly integrate this torsion element of M:

```
> AutonomousElements(R_mod, [x1(t),x2(t),u(t)], Alg)[2];
```

$$\left[\theta_1 = \_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}}) + \_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}}) \right]$$

The fact that there exists an autonomous element in the system is equivalent to the existence of a first integral of motion in the system. Indeed, let us recall that there exists a one-to-one correspondence between the torsion elements and the first integrals of motion (for more details, see J.-F. Pommaret, A. Quadrat, *Localization and parametrization of linear multidimensional control systems*, Systems & Control Letters, 37 (1999), pp. 247-260). We can compute this first integral of motion by using the command *FirstIntegral*:

```
> V := FirstIntegral(R_mod, [x1(t),x2(t),u(t)], Alg);
```

$$V := -(-(\frac{d}{dt}\,\mathrm{x1}(t))\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{l1} - (\frac{d}{dt}\,\mathrm{x1}(t))\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{l1}$$
$$+ \mathrm{x1}(t)\,\_C1 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{g} - \mathrm{x1}(t)\,\_C2 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{g}$$
$$+ (\frac{d}{dt}\,\mathrm{x2}(t))\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{l1} + (\frac{d}{dt}\,\mathrm{x2}(t))\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{l1}$$
$$- \mathrm{x2}(t)\,\_C1 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{g} + \mathrm{x2}(t)\,\_C2 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\sqrt{g})\,/\sqrt{l1}$$

Let us verify that the time derivative of V is zero modulo the system. Therefore, we first write the system in terms of the unknown functions $\mathrm{x1}(t)$, $\mathrm{x2}(t)$, $\mathrm{u}(t)$:

```
> S_mod := ApplyMatrix(R_mod, [x1(t),x2(t),u(t)], Alg);
```

$$S\_mod := \begin{bmatrix} \frac{g\,\mathrm{x1}(t)}{l1} + (\frac{d^2}{dt^2}\,\mathrm{x1}(t)) - \frac{g\,\mathrm{u}(t)}{l1} \\ \frac{g\,\mathrm{x2}(t)}{l1} + (\frac{d^2}{dt^2}\,\mathrm{x2}(t)) - \frac{g\,\mathrm{u}(t)}{l1} \end{bmatrix}$$

We find that the time derivative of V is a linear combination of the rows of S_mod:

```
> L := expand(evalm([coeff(diff(V, t), diff(x1(t),`$`(t,2))),
> -coeff(diff(V, t), diff(x1(t),`$`(t,2)))] &* S_mod)[1]);
```

$$L := \frac{\mathrm{x1}(t)\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\, g}{l1} + (\frac{d^2}{dt^2}\,\mathrm{x1}(t))\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}}) + \frac{\mathrm{x1}(t)\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\, g}{l1}$$
$$+ (\frac{d^2}{dt^2}\,\mathrm{x1}(t))\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}}) - \frac{\mathrm{x2}(t)\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})\, g}{l1} - (\frac{d^2}{dt^2}\,\mathrm{x2}(t))\,\_C1 \sin(\frac{\sqrt{g}\, t}{\sqrt{l1}})$$
$$- \frac{\mathrm{x2}(t)\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})\, g}{l1} - (\frac{d^2}{dt^2}\,\mathrm{x2}(t))\,\_C2 \cos(\frac{\sqrt{g}\, t}{\sqrt{l1}})$$

```
> simplify(diff(V, t)-L);
```

$$0$$

Finally, even if we have an autonomous element in the system, we can parametrize all solutions of the system in terms of one arbitrary function ξ_1 and two arbitrary constants $\_C1$, $\_C2$ these constants can easily be computed in terms of the initial conditions of the system):

```
> P := Parametrization(R_mod, Alg);
```

$$P := \left[\begin{array}{c} \_C1 \sin(\frac{\sqrt{g}\,t}{\sqrt{l1}}) + \_C2 \cos(\frac{\sqrt{g}\,t}{\sqrt{l1}}) + g\,\xi_1(t) \\ g\,\xi_1(t) \\ g\,\xi_1(t) + l1\,(\frac{d^2}{dt^2}\,\xi_1(t)) \end{array}\right]$$

We can easily check that P gives a parametrization of some solutions of the system as we have:

```
> simplify(ApplyMatrix(R_mod, P, Alg));
```

$$\left[\begin{array}{c} 0 \\ 0 \end{array}\right]$$

We can prove that we parametrize all the C^∞ solutions of the system. For more details, see A. Quadrat, D. Robertz, *On Monge problem for uncontrollable linear systems*, to appear.